$30



Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

MATHEMATICS

INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

CUBIC EQUATION

This program finds the roots of cubic equations of the form

$$f(x) = x^3+ax^2+bx+c=0$$

where a, b, and c are real.

It does so by extracting the first root, performing synthetic division, and solving the resulting quadratic equation (ref: HP-19C/HP-29C Applications Book, p.6).

Example 1:

$$x^3-6x^2+11x-6=0$$

Solution:

```
        CLRG
1.-04   STO0
-6.00   STO1
        STO3
11.00   STO2
        GSB1
3.00    ***     x1
        R/S
0.25    ***     D
        R/S
2.00    ***     x2
        R/S
1.00    ***     x3
```

Example 2:

$$x^3-4x^2+8x-8=0$$

Solution:

```
        CLRG
1.-04   STO0
-4.00   STO1
8.00    STO2
        CHS
        STO3
        GSB1
2.00    ***     x
        R/S
-3.00   ***     D
        GSB2
1.73    ***     v
        X⇄Y
1.00    ***     u
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		f	REG	
		10^{-4}	STO	0	
3.	Store coefficients	a	STO	1	
		b	STO	2	
		c	STO	3	
4.	Run		GSB	1	x_1
			R/S		D
5a	If D≥0, roots are real		R/S		x_2
			R/S		x_3
5b	If D<0 roots are u±iv		GSB	2	v
			x⇄y		u
6.	For a new case, go to step 2				

01 *LBL1		50 X>Y?	
02 RCL3		51 GTO7	
03 RCL3		52 RCL5	
04 ABS		53 STO4	
05 ÷		54 RCL8	
06 STO6		55 RCL6	
07 LSTX		56 x	
08 RCL1		57 X<0?	
09 ABS		58 GTO8	
10 +		59 GTO9	
11 RCL0		60 *LBL7	
12 *LBL0		61 RCL5	
13 1		62 R/S	*** x_1
14 0		63 RCL1	
15 x		64 +	
16 X≤Y?		65 CHS	
17 GTO0		66 STO1	
18 STO7		67 2	
19 *LBL9		68 ÷	
20 1		69 ENT↑	
21 0		70 X²	
22 ST÷7		71 RCL1	
23 RCL6		72 CHS	
24 CHS		73 RCL5	
25 STO6		74 x	
26 *LBL8		75 RCL2	
27 RCL7		76 +	
28 RCL6		77 STO3	
29 x		78 -	$D=(b^2-4ac)/4a^2$
30 RCL4		79 R/S	Real roots
31 +		80 √X	
32 STO5		81 X⇄Y	
33 RCL4		82 ABS	
34 X=Y?		83 +	
35 GTO7		84 RCL1	
36 RCL5		85 ENT↑	
37 RCL1		86 ABS	
38 +		87 ÷	
39 RCL5		88 x	*** x_2
40 x		89 R/S	
41 RCL2		90 RCL3	
42 +		91 X⇄Y	
43 RCL5		92 ÷	x_3
44 x		93 RTN	Immaginary roots
45 RCL3	f(x)	94 *LBL2	-D
46 +		95 CHS	u is in y register
47 STO8		96 √X	*** v
48 ABS	Tolerance	97 R/S	
49 RCL0			

REGISTERS

0 10^{-4}	1 a	2 b	3 c	4 Used	5 x
6 Used	7 Used	8 f(x)	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may replace "R/S"

SYNTHETIC DIVISION

This program performs synthetic division on a polynomial of degree n (with real coefficients)

$$a_n x^n + a_{n-1} x^{n-1} + \dots + a_1 x + a_0$$

by $x - x_0$ so that

$$a_n x^n + \dots + a_1 x + a_0$$
$$= (x - x_0)(b_{n-1} x^{n-1} + \dots + b_1 x + b_0) + R$$

where $b_{n-1} = a_n$

$$b_k = b_{k+1} x_0 + a_{k+1} \text{ for } k = n-2, \dots, 0$$

$$R = b_0 x_0 + a_0$$

Note: Program requires $n \le 7$. If $n < 7$, let $a_7 = \dots = a_{n+1} = 0$

Example:

$$(x^5 - 4x^4 + 7x^3 - 10x^2 + 8) \div (x-2)$$

Solution:

```
  0.00 ENT↑
  0.00 ENT↑
  1.00 ENT↑
 -4.00 GSB1
  7.00 ENT↑
-10.00 ENT↑
  0.00 ENT↑
  8.00  R/S
  2.00 GSB2
  0.00  ***   b6
        R/S
  0.00  ***   b5
        R/S
  1.00  ***   b4
        R/S
 -2.00  ***   b3
        R/S
  3.00  ***   b2
        R/S
 -4.00  ***   b1
        R/S
 -8.00  ***   b0
        R/S
 -8.00  ***   R
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store coefficients	a_7	ENT		
		a_6	ENT		
		a_5	ENT		
		a_4	GSB	1	a_7
		a_3	ENT		
		a_2	ENT		
		a_1	ENT		
		a_0	R/S		a_3
3.	Run	x_0	GSB	2	b_6
4.	Continue for i=5,4,...,0, R		R/S		b_i
5.	For a different x_0, go to step 3				

```
01 *LBL1
02  STO5
03    R↓
04  STO6
05    R↓
06  STO7
07    R↓
08  STO8
09   R/S
10  STO1
11    R↓
12  STO2
13    R↓
14  STO3
15    R↓
16  STO4
17   R/S
18 *LBL2
19  STO9
20     7
21  STO0
22  RCL8        *** b6
23   R/S
24 *LBL0
25  RCL9
26     x
27  RCLi
28     +        *** bi
29   R/S
30   DSZ
31  GTO0
32   R/S
```

REGISTERS

0 i	1 a_0	2 a_1	3 a_α	4 a_3	5 a_4
6 a_5	7 a_6	8 a_7	9 x_0	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

HYPERBOLIC FUNCTIONS
INVERSE HYPERBOLIC FUNCTIONS

This program calculates the hyperbolic functions and their inverses.

Equations:

$$\sinh x = \frac{e^x - e^{-x}}{2}$$

$$\cosh x = \frac{e^x + e^{-x}}{2}$$

$$\tanh x = \frac{\sinh x}{\cosh x}$$

$$\sinh^{-1} x = \ln[x + (x^2+1)^{1/2}]$$

$$\cosh^{-1} x = \ln[x + (x^2-1)^{1/2}] \quad (x \geq 1)$$

$$\tanh^{-1} x = \tfrac{1}{2}\ln[(1+x)/(1-x)] \quad (x^2 < 1)$$

$$\operatorname{csch} x = \frac{1}{\sinh x} \quad (x \neq 0)$$

$$\operatorname{sech} x = \frac{1}{\cosh x}$$

$$\coth x = \frac{1}{\tanh x} \quad (x \neq 0)$$

$$\operatorname{csch}^{-1} x = \sinh^{-1} \frac{1}{x} \quad (x \neq 0)$$

$$\operatorname{sech}^{-1} x = \cosh^{-1} \frac{1}{x} \quad (0 < x \leq 1)$$

$$\coth^{-1} x = \tanh^{-1} \frac{1}{x} \quad (x^2 > 1)$$

Examples:

1. sinh (1.5)
2. cosh (5.9)
3. tanh (1.3)
4. csch (.95)
5. sech (-3)
6. coth (-1.99)
7. $\sinh^{-1}$ (3.5)
8. $\cosh^{-1}$ (100)
9. $\tanh^{-1}$ (-.7)
10. csch^{-1} (3)
11. sech^{-1} (.5)
12. $\coth^{-1}$ (5.4)

Solutions:

```
1.     1.50 GSB1
       2.13  ***
2.     5.90 GSB2
     182.52  ***
3.     1.30 GSB3
       0.86  ***
4.     0.95 GSB1
            GSB4
       0.91  ***
5.    -3.00 GSB2
            GSB4
       0.10  ***
6.    -1.99 GSB3
            GSB4
      -1.04  ***
7.     3.50 GSB5
       1.97  ***
8.   100.00 GSB6
       5.30  ***
9.    -0.70 GSB7
      -0.87  ***
10.    3.00 GSB4
            GSB5
       0.33  ***
11.    0.50 GSB4
            GSB6
       1.32  ***
12.    5.40 GSB4
            GSB7
       0.19  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	sinh x	x	GSB	1	sinh x
3.	cosh x	x	GSB	2	cosh x
4.	tanh x	x	GSB	3	tanh x
5.	csch x	x	GSB	1	
			GSB	4	csch x
6.	sech x	x	GSB	2	
			GSB	4	sech x
7.	coth x	x	GSB	3	
			GSB	4	coth x
8.	$\sinh^{-1}x$	x	GSB	5	$\sinh^{-1} x$
9.	$\cosh^{-1}x$	x	GSB	6	$\cosh^{-1} x$
10.	$\tanh^{-1}x$	x	GSB	7	$\tanh^{-1} x$
11.	$\text{csch}^{-1}x$	x	GSB	4	
			GSB	5	$\text{csch}^{-1} x$
12.	$\text{sech}^{-1}x$	x	GSB	4	
			GSB	6	$\text{sech}^{-1} x$
13.	$\coth^{-1}x$	x	GSB	4	
			GSB	7	$\coth^{-1} x$

```
01 *LBL1
02    e^x
03  ENT↑
04   1/X
05     -
06     2
07     ÷
08   RTN          *** sinh x
09 *LBL2
10    e^x
11  ENT↑
12   1/X
13     +
14     2
15     ÷
16   RTN          ***  cosh x
17 *LBL3
18    e^x
19  STO1
20  ENT↑
21   1/X
22     -
23  RCL1
24  LSTX
25     +
26     ÷
27   RTN          ***  tanh x
28 *LBL4
29   1/X
30   RTN          ***
31 *LBL5
32  ENT↑
33    X²
34     1
35     +
36    √X
37     +
38    LN
39   RTN          *** sinh⁻¹x
40 *LBL6
41  ENT↑
42    X²
43     1
44     -
45    √X
46     +
47    LN
48   RTN
49 *LBL7          *** cosh⁻¹x
50  ENT↑
51     1
52   X⇄Y
53     +
54     1
55  LSTX
56     -
57     ÷
58    LN
59     2
60     ÷
61   RTN
62   R/S          *** tanh⁻¹x
```

REGISTERS

0	1 Used	2	3	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may be inserted.

POLYNOMIAL EVALUATION--REAL OR COMPLEX

This program evaluates polynomials of the form:

$f(x_0) = a_0+a_1x+\ldots+a_nx^n$

where the coefficients

$a_k, k=0,\ldots n (n\leq 28)$ and x_0 are real or the coefficients and x_0 are complex of the form

$a_k = Re(a_k) + i\ Im(a_k)$

$z_0 = Re(z_0) + i\ Im(z_0)$

$k = 0, 1, \ldots n$

Example 1:

$f(x) = 11-7x - 3x^2 + 5x^4 + x^5$

for $x_0 = 2.5$

for $x_0 = -5$

Solution:

```
        CLRG
 11.00  GSB1
 -7.00  R/S
 -3.00  R/S
  0.00  R/S
  5.00  R/S
  1.00  R/S
  2.50  GSB2
267.72  ***    f (2.5)
  6.00  STO0
 -5.00  GSB2
-29.00  ***    f (-5)
```

Example 2:

$f(x) = (5-7i) - 10x + (-2+i)x^2 + 18x^3 + (3+4i)x^4$

for $x_0 = 2 + i$

Solution:

```
   1.00  ENT↑
   2.00  GSB3
   4.00  ENT↑
   3.00  GSB4
   0.00  ENT↑
  18.00  GSB4
   1.00  ENT↑
  -2.00  GSB4
   0.00  ENT↑
 -10.00  GSB4
  -7.00  ENT↑
   5.00  GSB5
-106.00  ***   Re f(x0)
         R/S
 220.00  ***   Im f(x0)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
	For real polynomials:				
2.	Initialize		f	REG	
3.	Store coefficients	a_0	GSB	1	
4.	Continue for $i = 1$ to n, $n \leq 28$	a_i	R/S		
5.	Compute $f(x_0)$	x_0	GSB	2	$f(x_0)$
6.	For a different x_0 go to step 5	n+1	STO	0	
	For complex polynomials:				
7.	Enter x_0	Im x_0	ENT		
		Re x_0	GSB	3	0
8.	Enter a_k, $k=n,n-1,\ldots 1$	Im a_k	ENT		
		Re a_k	GSB	4	
9.	Enter a_0 and run	Im a_0	ENT		
		Re a_0	GSB	5	Re $f(x_0)$
			R/S		Im $f(x_0)$
10.	For a new case, go to step 7				

01 *LBL1		50 *LBL9	Add real parts and
02 ISZ		51 X⇄Y	imaginary parts
03 STOi		52 R↓	
04 R/S		53 +	
05 GTO1		54 R↓	
06 *LBL2		55 +	
07 ENT↑		56 X⇄Y	
08 ENT↑		57 R↓	
09 ENT↑		58 X⇄Y	
10 RCLi		59 RTN	
11 ×	Multiply by x_0	60 R/S	
12 DSZ			
13 *LBL0			
14 RCLi			
15 +			
16 ×	Multiply by x_0		
17 DSZ			
18 GTO0			
19 X⇄Y			
20 ÷	Divide by x_0		
21 R/S	*** $f(x_0)$		
22 *LBL3	Routines for		
23 →P	complex polynomials		
24 STO1	r		
25 X⇄Y			
26 STO2	θ		
27 0			
28 ENT↑	Prepare for LBL 9		
29 ENT↑			
30 ENT↑			
31 RTN			
32 *LBL4			
33 GSB9			
34 GTO8			
35 *LBL5			
36 GSB9			
37 R/S	*** Re $f(x_0)$		
38 X⇄Y			
39 R/S	*** Im $f(x_0)$		
40 *LBL8			
41 →P	Multiply in polar		
42 RCL1	form		
43 ×			
44 X⇄Y			
45 RCL2			
46 +			
47 X⇄Y			
48 →R			
49 RTN			

REGISTERS

0 i	1 r or a_0	2 θ or a_1	3 a_2	4 . . . a_n	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29 a_{28}

*** "Printx" may be inserted or used to replace "R/S".

ROOTS OF F(X) = 0 IN AN INTERVAL

This program uses a half-interval search to find the real roots of an equation $f(x) = 0$ in a closed interval $[a,b]$.

The user specifies the continuous, real function f, an interval $[a,b]$, an accuracy tolerance ε, and a search increment Δx. The program then begins at the left of the interval and compares the functional values at the ends of the interval $[a, a + \Delta x]$. If $f(a)$ and $f(a + \Delta x)$ are of opposite sign, this interval is searched for a root. Otherwise, or even after a root is found, the program proceeds in the same manner with the interval $[a + \Delta x, a + 2\Delta x]$, etc. At most one root will be found by the program for each of these small intervals.

33 lines and 22 registers ($R_0, R_9 - R_{29}$) available for defining $f(x)$

<u>Example 1</u>:

Find the roots of $x^3-8x^2+5x+14 = 0$ in the interval $[-10,10]$ using $\Delta x = 1$ and $\varepsilon = 10^{-6}$

<u>Solution</u>:

```
          GTO0
66  STO0
67    3
68    Yx
69  RCL0
70    X²
71    8
72    x         f(x)
73    -
74    5
75  RCL0
76    x
77    +
78    1
79    4
80    +
81  RTN
```

```
 1.-06  STO5
  1.00  STO6
-10.00  STO1
 10.00  STO7
        GSB1
 -1.00  ***    x1
        R/S
  2.00  ***    x2
        R/S
  7.00  ***    x3
        R/S
 11.00  ***    b+Δx
```

<u>Example 2</u>:

Find the root of $x^{5/2} - 2\sqrt{x} = 0$ in the interval $[1,10]$ using $\Delta x = 1$ and $\varepsilon = 10^{-6}$

<u>Solution</u>:

```
          GTO0
66   √X
67  ENT↑
68  ENT↑
69    5
70    Yx           f(x)
71   X⇄Y
72    2
73    x
74    -
75   RTN
      1.-06  STO5
       1.00  STO6
      10.00  STO7
       1.00  STO1
             GSB1   x1
       1.41  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Enter program with f(x) defined at LBL 0				
	and ending with RTN				
2.	Store constants	ε	STO	5	
		Δx	STO	6	
		b	STO	7	
		a	STO	1	
3.	Run		GSB	1	root
4.	Continue until display = b + Δx indicating the end of search		R/S		root or b + Δx
5.	For a different f(x): press GTO 0; switch to program mode; key in f(x) ending with g RTN; switch to run mode; and go to step 3.				

01	*LBL1	
02	RCL1	x
03	GSB0	
04	STO3	f(x)
05	X=0?	
06	GSB9	
07	RCL1	
08	RCL6	
09	+	
10	STO2	
11	STO8	
12	GSB0	
13	RCL3	
14	x	
15	X<0?	
16	GTO8	
17	RCL2	
18	STO1	
19	RCL6	
20	+	
21	STO2	
22	RCL7	
23	X⇄Y	
24	X>Y?	
25	R/S	End of interval
26	GTO1	
27	*LBL6	
28	RCL4	
29	STO2	
30	GTO8	
31	*LBL7	
32	RCL4	
33	R/S	*** root
34	RCL8	
35	STO1	
36	GTO1	
37	*LBL9	
38	RCL1	
39	R/S	*** root
40	RTN	
41	*LBL8	
42	RCL1	
43	RCL2	
44	+	
45	2	
46	÷	Halve interval
47	STO4	
48	GSB0	
49	ABS	
50	RCL5	
51	X>Y?	f(x)<tolerance?
52	GTO7	
53	RCL1	
54	GSB0	
55	STO3	
56	RCL4	
57	GSB0	
58	RCL3	
59	x	
60	X<0?	Sign change?
61	GTO6	
62	RCL4	
63	STO1	
64	GTO8	
65	*LBL0	x . . . . f(x)

REGISTERS

0	1 a	2 Used	3 f(x)	4 Used	5 ε
6 Δx	7 b	8 Used	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may be inserted

3 x 3 MATRIX INVERSION

$$A = \begin{pmatrix} a_1 & b_1 & c_1 \\ a_2 & b_2 & c_2 \\ a_3 & b_3 & c_3 \end{pmatrix}$$

has an inverse

$$A^{-1} = \begin{pmatrix} \alpha_1 & \alpha_4 & \alpha_7 \\ \alpha_2 & \alpha_5 & \alpha_8 \\ \alpha_3 & \alpha_6 & \alpha_9 \end{pmatrix}$$

where $\alpha_1 = (b_2c_3-b_3c_2)/\mathrm{Det}$

$\alpha_2 = (a_3c_2-a_2c_3)/\mathrm{Det}$

$\alpha_3 = (a_2b_3-a_3b_2)/\mathrm{Det}$

$\alpha_4 = (b_3c_1-b_1c_3)/\mathrm{Det}$

$\alpha_5 = (a_1c_3-a_3c_1)/\mathrm{Det}$

$\alpha_6 = (a_3b_1-a_1b_3)/\mathrm{Det}$

$\alpha_7 = (b_1c_2-b_2c_1)/\mathrm{Det}$

$\alpha_8 = (a_2c_1-a_1c_2)/\mathrm{Det}$

$\alpha_9 = (a_1b_2-a_2b_1)/\mathrm{Det}$

if the determinant Det of A is non-zero.

Example:

$$A = \begin{pmatrix} -1 & 0 & 3 \\ 7 & 1 & -1 \\ 2 & 3 & 0 \end{pmatrix}$$

Solution:

```
-1.00 ST01
 7.00 ST02
 2.00 ST03
 0.00 ST04
 1.00 ST05
 3.00 ST06
 3.00 ST07
-1.00 ST08
 0.00 ST09
      GSB1
54.00  ***   Det
       R/S
 0.06  ***   α1
       R/S
-0.04  ***   α2
       R/S
 0.35  ***   α3
       R/S
 0.17  ***   α4
       R/S
-0.11  ***   α5
       R/S
 0.06  ***   α6
       R/S
-0.06  ***   α7
       R/S
 0.37  ***   α8
       R/S
-0.02  ***   α9
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store matrix values	a_1	STO	1	
		a_2	STO	2	
		a_3	STO	3	
		b_1	STO	4	
		b_2	STO	5	
		b_3	STO	6	
		c_1	STO	7	
		c_2	STO	8	
		c_3	STO	9	
3.	Run		GSB	1	Det.
4.	For k = 1, 2, ... 9		R/S		α_k

01 *LBL0		50 RCL2	
02 x		51 GSB9	
03 x		52 RCL9	
04 RTN		53 RCL4	
05 *LBL1		54 RCL7	
06 RCL1		55 RCL6	
07 RCL5		56 GSB9	
08 RCL9		57 RCL7	
09 GSB0		58 RCL3	
10 RCL2		59 RCL9	
11 RCL6		60 RCL1	
12 RCL7		61 GSB9	
13 GSB0		62 RCL6	
14 +		63 RCL1	
15 RCL3		64 RCL3	
16 RCL4		65 RCL4	
17 RCL8		66 GSB9	
18 GSB0		67 RCL7	
19 +		68 RCL5	
20 RCL3		69 RCL8	
21 RCL5		70 RCL4	
22 RCL7		71 GSB9	
23 GSB0		72 RCL8	
24 -		73 RCL1	
25 RCL2		74 RCL7	
26 RCL4		75 RCL2	
27 RCL9		76 GSB9	
28 GSB0		77 RCL4	
29 -		78 RCL2	
30 RCL1		79 RCL5	
31 RCL6		80 RCL1	
32 RCL8		81 *LBL9	
33 GSB0		82 x	
34 -		83 R↓	
35 R/S	Determinant	84 x	
36 ST.0		85 X⇄Y	
37 RCL8		86 R↓	
38 RCL6		87 -	
39 RCL9		88 RC.0	
40 RCL5		89 ÷	
41 GSB9		90 R/S	*** α_k
42 RCL9		91 RTN	
43 RCL2			
44 RCL8			
45 RCL3			
46 GSB9			
47 RCL5			
48 RCL3			
49 RCL6			

REGISTERS

0	1 a_1	2 a_2	3 a_3	4 b_1	5 b_2
6 b_3	7 c_1	8 c_2	9 c_3	.0 D	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may be used to replace "R/S".

BASE b ARITHMETIC

Given positive integers x and y, this program computes $y_b \circ x_b$ where the operation $\circ$ can be +, -, x, ÷. In division, the remainder is truncated.

The program can also be used to perform base conversions. When the base has two digits, two digits are allocated in the display. For example $2BD7_{16}$ would appear as 2121407.

Reference:

"Applications Programs, Volume 2", Adams, ed. Int'l. Software Clearinghouse, Estacada, Oregon, 1977.

Examples:

1. $213_8 + 37507_8$
2. $12_8 - 37_8$
3. $12345_8 \times 4567_8$
4. $16_8 \div 4_8$
5. $A3C9_{16} \rightarrow Base_{10}$
6. $30690_{10} \rightarrow$ Base 16

Solutions:

```
       8. STO7
     213. ENT↑
   37507. GSB1
   37722.  ***
      12. ENT↑
      37. GSB2
     -25.  ***
   12345. ENT↑
    4567. GSB3
61341563.  ***
      16. ENT↑
       4. GSB4
       3.  ***
      16. STO7
-10031209. GSB0
   41929.  ***
   30690. GSB0
          CHS
 7071402.  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Initialize display		f	fix	
			0		
2.	Store base	b	STO	7	
3.	y + x	y	ENT		
		x	GSB	1	y + x
4.	y - x	y	ENT		
		x	GSB	2	y - x
5.	y • x	y	ENT		
		x	GSB	3	y • x
6.	y ÷ x Note: remainder is truncated	y	ENT		
		x	GSB	4	y ÷ x
	For base conversions:				
1.	Store base	b	STO	7	
2.	To convert an integer in base b to base $_{10}$	N_b	CHS		
			GSB	0	N_{10}
3.	To convert an integer in base 10 to base b	N_{10}	GSB	0	
			CHS		N_b

Step	Key	Comment	Step	Key
01	*LBL9		50	RCL7
02	CHS	-x	51	LOG
03	GSB0		52	INT
04	X⇄Y		53	1
05	CHS	-y	54	STO3
06	GSB0		55	+
07	RTN		56	10^x
08	*LBL1		57	STO1
09	GSB9	x_b y_b	58	RCL4
10	+	y_{10} x_{10}	59	X<0?
11	GSB0	→ Base b	60	GTO8
12	CHS		61	R↓
13	R/S	*** y + x	62	STO5
14	*LBL2		63	RCL7
15	GSB9	x_b y_b	64	STO1
16	X⇄Y	y_{10} x_{10}	65	*LBL8
17	-	y-x	66	RCL4
18	X<0?		67	*LBL7
19	GTO6		68	RCL1
20	GSB0	→ Base b	69	÷
21	CHS	*** y - x	70	STO6
22	R/S		71	FRC
23	*LBL6		72	RCL1
24	CHS		73	x
25	GSB0		74	RCL3
26	R/S		75	x
27	*LBL3		76	EEX
28	GSB9		77	3
29	x		78	+
30	GSB0		79	LSTX
31	CHS		80	-
32	R/S	*** y · x	81	ST+2
33	*LBL4		82	RCL5
34	GSB9		83	ST×3
35	X⇄Y		84	RCL6
36	÷		85	INT
37	GSB0		86	X≠0?
38	CHS	*** y ÷ x	87	GTO7
39	R/S		88	RCL0
40	*LBL0	Base b to base 10	89	RCL2
41	INT	Base 10 to base b	90	CHS
42	RCL7	conversion routine	91	RTN
43	CLRG			
44	STO5	↓		
45	STO7			
46	R↓			
47	STO4			
48	R↓			
49	STO0			

REGISTERS

0 Used	1 Used	2 Used	3 Used	4 Used	5 Used
6 Used	7 Used	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

GAUSSIAN QUADRATURE $\int_a^b F(x)\,dx$

This program computes the value of

$$\int_a^b f(x)\,dx$$

for finite a and b, and an f(x) which is single-valued in the range [a,b], using the six point Gauss–Legendre quadrature formula:

$$\int_a^b f(x) \approx$$

$$\frac{b-a}{2} \sum_{i=1}^{6} w_i\, f\left(\frac{z_i\,(b-a) + b + a}{2}\right)$$

where:

$$z_1 = -z_2 = .238619186$$
$$z_3 = -z_4 = .661209386$$
$$z_5 = -z_6 = .932469514$$
$$w_1 = w_2 = .467913935$$
$$w_3 = w_4 = .360761573$$
$$w_5 = w_6 = .171324492$$

33 lines and 20 registers ($R_{.0}$-R_{29}) are available to define f(x).

Reference: Applied Numerical Methods. Carnahan, Luther, and Wilks; John Wiley and Sons, 1969.

Examples:

1. $\int_1^{10} \frac{dx}{x}$

2. $\int_e^{e^2} \frac{dx}{x(\ln x)^3}$

Solutions:

```
0.238619186 ST01
0.661209386 ST02
0.932469514 ST03
0.467913935 ST04
0.360761573 ST05
0.171324492 ST06
            GT02
66   1/X              f(x)
67   RTN
       1.00 ENT↑
      10.00 GSB1
       2.30 ***
            FIX8
 2.30140808 ***

            GT02
66  ENT↑
67    LN
68     3              f(x)
69    Yˣ
70     x
71   1/X
72   RTN
 1.00000000   eˣ
            ENT↑
              X²
            GSB1
 0.37497974  ***
```

The correct answers are:

1. ln 10
2. 3/8

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program and switch to run mode				
2.	Store constants:	.238619186	STO	1	
		.661209386	STO	2	
		.932469514	STO	3	
		.467913935	STO	4	
		.360761573	STO	5	
		.171324492	STO	6	
3.	Go to Label 2		GTO	2	
4.	Switch to program mode and key in f(x).				
	Include a RTN statement .				
5.	Switch to run mode				
6.	Key in limits and run	a	ENT↑		
		b	GSB	1	$\int f(x)$
7.	For a new case, go to step 3.				

```
01 *LBL1
02  STO8
03  STO7
04    R↓
05  ST+7
06  ST-8
07     0
08  STO9
09     1
10   CHS
11  STO0        Clear flag
12 *LBL0
13  GSB9
14  RCL1
15  RCL2
16  STO1
17   X⇄Y
18  STO2
19  RCL4
20  RCL5
21  STO4
22   X⇄Y
23  STO5
24  GSB9
25  RCL1
26  RCL3
27  STO1
28   X⇄Y
29  STO3
30  RCL4
31  RCL6
32  STO4
33   X⇄Y
34  STO6
35  GSB9
36  RCL0        Test flag
37  X>0?
38  GTO8
39 *LBL6        -1
40  ST×1
41  ST×2
42  ST×3
43  ST×0        Set flag
44  GTO0
45 *LBL8
46  RCL9
47  RCL8
48     ×
49     2
50     ÷        *** ∫_a^b f(x) dx
51   R/S
52 *LBL9
53  RCL1
54  RCL8
55     ×
56  RCL7
57     +
58     2
59     ÷
60  GSB2
61  RCL4
62     ×
63  ST+9
64   RTN
65 *LBL2        f(x)
66   R/S
```

REGISTERS

0 Flag	1 z_1 (z_2)	2 z_3 (z_4)	3 z_5 (z_6)	4 w_1 (w_2)	5 w_3 (w_4)
6 w_5 (w_6)	7 a + b	8 b − a	9 Used	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

GAUSSIAN QUADRATURE FOR $\int_a^\infty F(x)\,dx$

This program computes the value $\int_a^\infty f(x)\,dx$ for finite a and single valued function f(x) by the six point Gauss-Legendre quadrature formula:

$$\int_a^\infty f(x)\,dx \approx \frac{1}{2}\sum_{i=1}^{6} \frac{4\,w_i}{(1+z_i)^2}\, f\left(\frac{2}{1+z_i} + a - 1\right)$$

where: $z_1 = -z_2 = .238619186$

$z_3 = -z_4 = .661209386$

$z_5 = -z_6 = .932469514$

$w_1 = w_2 = .467913935$

$w_3 = w_4 = .360761573$

$w_5 = w_6 = .171324492$

33 lines and 20 registers ($R_{.0}$ - R_{29}) are available to define f(x).

Reference: Applied Numerical Methods. Carnahan, Luther, and Wilks, John Wiley and Sons, 1969.

Examples:

1. $\int_0^\infty e^{-x}\, x^{.8}\, dx$

2. $\int_0^\infty \frac{dx}{(x^2+1)(x^2+4)^2}$

Solutions:

```
0.238619186 ST01
0.661209386 ST02
0.932469514 ST03
0.467913935 ST04
0.360761573 ST05
0.171324492 ST06

                 GT02
66   CHS
67    eˣ
68  LSTX
69   CHS                 f(x)
70    .
71    8
72   Yˣ
73    x
74   RTN

                 FIX8
  0.00000000 ENT↑
                 GSB1
  0.92410105  ***

                 GT02
66   X²
67   1
68   +
69  ENT↑
70  ENT↑                 f(x)
71   3
72   +
73   X²
74   x
75  1/X
76  RTN
  0.00000000 ENT↑
                 GSB1
  0.05453121  ***
```

The correct answers are:

1. $\Gamma(1.8) = .931383771$
2. $5\pi/288$

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program and switch to run mode.				
2.	Store constants:	.238619186	STO	1	
		.661209386	STO	2	
		.932469514	STO	3	
		.467913935	STO	4	
		.360761573	STO	5	
		.171324492	STO	6	
3.	Go to Label 2		GTO	2	
4.	Switch to program mode and key in f(x). Include a RTN statement.				
5.	Switch to run mode				
6.	Key in a and run	a	GSB	1	$\int f(x)$
7.	For a new case, go to step 3.				

Step	Comment
01 *LBL8	
02 RCL8	
03 2	
04 ÷	
05 R/S	*** $\int f(x)$
06 *LBL1	
07 ENT↑	
08 1	
09 CHS	
10 STO0	Clear flag
11 +	
12 STO7	
13 0	
14 STO8	
15 *LBL0	
16 GSB9	
17 RCL1	
18 RCL2	
19 STO1	
20 X⇄Y	
21 STO2	
22 RCL4	
23 RCL5	
24 STO4	
25 X⇄Y	
26 STO5	
27 GSB9	
28 RCL1	
29 RCL3	
30 STO1	
31 X⇄Y	
32 STO3	
33 RCL4	
34 RCL6	
35 STO4	
36 X⇄Y	
37 STO6	
38 GSB9	
39 RCL0	
40 X>0?	Test flag
41 GTO8	
42 *LBL6	-1
43 ST×0	
44 ST×1	Set flag
45 ST×2	
46 ST×3	
47 GTO0	
48 *LBL9	
49 2	
50 RCL1	
51 1	
52 +	
53 ÷	
54 STO9	
55 RCL7	
56 +	
57 GSB2	
58 RCL9	
59 X²	
60 ×	
61 RCL4	
62 ×	
63 ST+8	
64 RTN	
65 *LBL2	$f(x)$

REGISTERS

0 flag	1 z_1 (z_2)	2 z_3 (z_4)	3 z_5 (z_6)	4 w_1 (w_2)	5 w_3 (w_4)
6 w_5 (w_6)	7 a - 1	8 Used	9 Used	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

BESSEL FUNCTION J (X)

This program computes the value of the Bessel function $J_n(X)$ by using a numerical method which makes use of the recurrence relation

$$J_{n-1}(X) = \frac{2n}{X} J_n(X) - J_{n+1}(X)$$

the summation relation

$$J_0(X) + 2 \sum_{i=1}^{\infty} J_{2i}(X) = 1$$

and the fact that

$$\lim_{n \to \infty} J_n(X) = 0$$

First let

$$m = \text{INT} \left\{ 1 + 3x^{\frac{1}{12}} + 9x^{\frac{1}{3}} + \max(n,x) \right\}$$

where INT means "integer part of".

Then set

$$T_m = a \qquad T_{m+1} = 0$$

where a is an arbitrary non-zero constant.

Then the series of terms, $T_k, 0 \le k \le m$, is computed by successively applying the relation.

$$T_{k-1}(X) = \frac{2k}{X} T_k(X) - T_{k+1}(X)$$

starting with k = m.

$J_n(X)$ is then found by dividing the term $T_n(X)$ by the normalizing constant

$$K = T_0(X) + 2 \sum_{i=1}^{p} T_{2i}(X)$$

where

$$p = \begin{cases} \frac{m}{2} & \text{if m is even} \\ \frac{m-1}{2} & \text{if m is odd} \end{cases}$$

Note that all the T_k are proportional to a, hence K and the result are independent of a.

Note: $J_0(X) = 1$ for $x \le 10^{-6}$ but is out of range for this program.

Examples:

1. J_0 (4.7)
2. J_5 (9.2)
3. J_0 (1)
4. J_5 (5)

Solutions:

```
 0.00000000 ENT↑
 4.70000000 GSB1
-0.26933079  ***
 5.00000000 ENT↑
 9.20000000 GSB1
-0.10052862  ***
 0.00000000 ENT↑
 1.00000000 GSB1
 0.76519769  ***
 5.00000000 ENT↑
 5.00000000 GSB1
 0.26114055  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Compute $J_n(X)$	n	ENT		
		X	GSB	1	$J_n(X)$

Step		Step	
01 *LBL1		50 RCL1	
02 ST01		51 ÷	
03 R↓		52 x	
04 ST05		53 RCL6	
05 EEX		54 ST03	
06 CHS		55 x	
07 9		56 X⇄Y	
08 9		57 -	
09 ST06		58 ST06	
10 0		59 RCL8	
11 ST03		60 1	Decrement count
12 ST04		61 -	
13 RCL1	Calculate m	62 GTO9	
14 6		63 *LBL7	
15 1/X		64 RCL7	
16 Y^X		65 RCL4	
17 ENT↑		66 2	
18 ENT↑		67 x	
19 9		68 RCL6	
20 x		69 +	
21 x		70 ÷	
22 LSTX		71 R/S	*** $J_n(X)$
23 √X		72 *LBL0	
24 +		73 RCL6	
25 1		74 ST07	
26 +		75 RTN	
27 RCL1		76 *LBL8	
28 RCL5		77 RCL6	
29 X>Y?		78 ST+4	
30 X⇄Y		79 RTN	
31 R↓	MAX(n or x)		
32 +			
33 INT			
34 *LBL9	m		
35 ST08			
36 RCL5			
37 X=Y?			
38 GSB0			
39 RCL8			
40 X=0?			
41 GTO7			
42 2			
43 ÷			
44 FRC			
45 X=0?			
46 GSB8			
47 RCL3			
48 RCL8			
49 2			

REGISTERS

0	1 X	2	3 T_{k+1}	4 ΣT_{2i}	5 n
6 $10^{-99}, T_k$	7 Tn	8 Counter k	9 Used	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may be inserted.

GAMMA FUNCTION

This program approximates the gamma function $\Gamma(x)$ for $0<x\leq 61$ with eight digit accuracy over most of the range.

Equations:

(1) $\Gamma(x) = e^{\left[\ln\sqrt{\frac{2\pi}{x}} + x \ln x - x + A\right]}$

where $A = (1 - \frac{1}{30x^2} + \frac{1}{105x^4}) \cdot \frac{1}{12x}$

(2) $\Gamma(x+1) = x\Gamma(x)$

Note: This program can be used to find $x! = \Gamma(x+1)$

Examples:

1. $\Gamma(1)$
2. $\Gamma(.5)$
3. $\Gamma(5.25)$
4. 7!

Solutions:

```
   1.00000000 GSB1
   1.00000000  ***   Γ(1)
   0.50000000 GSB1
   1.77245385  ***   Γ(.5)=√π
   5.25000000 GSB1
  35.21161167  ***   Γ(5.25)
   8.00000000 GSB1
 5040.000017  ***   7!=Γ(8)
```

Reference: Gamma Function, John Ulissides. "65 Notes," V 3 N 10, p. 37.

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS	OUTPUT DATA/UNITS
1.	Key in the program			
2.	Compute $\Gamma(x)$	x	GSB 1	$\Gamma(x)$

```
01 *LBL1
02  STO0
03    9
04    +          T = x + 9
05  ENT↑
06   1/X
07    X²
08  ENT↑
09    X²
10    3
11    .
12    5
13    ÷
14    -
15    3
16    0
17    ÷
18    1
19    -
20   X⇄Y
21    1
22    2
23    x
24    ÷
25    +
26   X⇄Y
27  ENT↑
28    LN
29    x
30    -
31   X⇄Y
32    PI
33    ÷
34    2
35    ÷
36    √X
37    LN
38    +          ln Γ (T)
39   CHS
40    eˣ
41  STO1
42   CLX
43    9          x
44    -          Reduce Γ (T)
45 *LBL0
46  ST÷1
47    1
48    +          x + i   x + 9
49  X≠Y?
```

```
50  GTO0
51  RCL1
52   R/S         *** Γ (x)
```

REGISTERS

0 x	1 Γ(x)	2	3	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Printx" may replace "R/S".

SINE, COSINE, AND EXPONENTIAL INTEGRALS

Sine integral:

$$Si(x) = \int_0^x \frac{\sin t}{t}\, dt = \sum_{n=0}^{\infty} \frac{(-1)^n x^{2n+1}}{(2n+1)(2n+1)!}$$

where x is real.

This routine computes successive partial sums of the series, stops when two consecutive partial sums are equal, and displays the last partial sum as the answer.



Notes: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

$$Si(-x) = -Si(x)$$

Cosine integral:

$$Ci(x) = \gamma + \ln x + \int_0^x \frac{\cos t - 1}{t}\, dt = \gamma + \ln x + \sum_{n=1}^{\infty} \frac{(-1)^n x^{2n}}{2n(2n)!}$$

where $x>0$, and $\gamma = 0.577215665$ is the Euler's constant.

This program computes successive partial sums of the series. When two consecutive partial sums are equal, the value is used as the sum of the series.



Notes: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

$$Ci(-x) = Ci(x) - i\pi \text{ for } x>0.$$

Exponential integral:

$$Ei(x) = \int_{-\infty}^{x} \frac{e^t}{t}\, dt = \gamma + \ln x + \sum_{n=1}^{\infty} \frac{x^n}{nn!}$$

where $x>0$ and $\gamma = 0.577215665$ is Euler's constant.

This program computes successive partial sums of the series. When two consecutive partial sums are equal, the value is used as the sum of the series.



Note: When x is too large, computing a new term of the series might cause an overflow. In that case, display shows all 9's and the program stops.

References: Handbook of Mathematical Functions, Abramowitz and Stegun, National Bureau of Standards, 1968.

Examples:

1. Si (.69)
2. Si (.98)
3. Ci (1.38)
4. Ci (5)
5. Ei (1.59)
6. Ei (.61)

Solutions:

```
0.577215665 ST.0
       0.69 GSB1
       0.67  ***
       0.98 GSB1
       0.93  ***
       1.38 GSB2
       0.46  ***
       5.00 GSB2
      -0.19  ***
       1.59 GSB3
       3.57  ***
       0.61 GSB3
       0.80  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store α	.577215665	STO	•	
			0		α
3.	For Sine Integral	x	GSB	1	Si(x)
	For Cosine Integral	x	GSB	2	Ci(x)
	For Exponential Integral	x	GSB	3	Ei(x)

```
01 *LBL1       Sine Integral
02  STO3       routine
03    X²
04   CHS
05  STO1
06     1
07  STO2
08  RCL3
09 *LBL0
10  RCL1
11  RCL2
12     1
13     +
14     ÷
15  LSTX
16     1
17     +
18  STO2
19     ÷
20  RCL3
21     x
22  STO3
23  RCL2
24     ÷
25     +
26  X≠Y?
27  GTO0
28   R/S       *** Si(x)/Ci(x)
29 *LBL2       Cosine Integral
30    X²       routine
31   CHS
32  STO1
33     1
34  STO3
35     0
36  STO2
37  LSTX
38    LN
39  RC.0
40     +
41  GTO0
42 *LBL3
43  STO1       Exponential In-
44     1       tegral routine
45  STO3
46     0
47  STO2
48  RCL1
49    LN
50  RC.0
51     +
52 *LBL9
53  RCL1
54  RCL2
55     1
56     +
57  STO2
58     ÷
59  RCL3
60     x
61  STO3
62  RCL2
63     ÷
64     +
65  X≠Y?
66  GTO9
67   R/S       *** Ei(x)
```

REGISTERS

0	1 used	2 used	3 used	4	5
6	7	8	9	.0 used	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

*** "Print X" may be used to replace "R/S".

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions